Réduction des courbes du 2nd degré

Jean-Michel Sarlat
17 décembre 2003

Ce document a été généré par la ***fabrique de Syracuse*** *à partir d'un fichier texte contenant des spécifications de contenu et des commandes d'interfaçage avec* ***Maxima****. Le traitement de ce fichier a donné lieu, dans le même temps que ce document, à la création d'une page* html.
URL : http ://melusine.eu.org/syracuse/maxima/

1 Présentation

L'objectif est ici de déterminer la nature des courbes dont l'équation dans un plan euclidien rapporté à un repère orthonormé est de la forme :

$$\alpha x^2 + \beta xy + \gamma y^2 + ax + by + c = 0$$

avec $\alpha^2 + \beta^2 + \gamma^2 \neq 0$. Ces courbes sont des coniques éventuellement *dégénérées*.

2 Mise en place de la procédure

Soit $\mathscr{C} : 16x^2 - 24xy + 9y^2 + 35x - 20y = 0$ à réduire. Dans un premier temps nous allons effectuer une rotation de repère pour éliminer le terme rectangle (en xy).

2.1 Élimination du terme rectangle

maxima >>

```
(C2) eq1:16*x^2-24*x*y+9*y^2+35*x-20*y$
```

On substitue à x et y leurs expressions en fonction des coordonnées X et Y dans le nouveau repère obtenu à l'aide d'une rotation d'angle φ.

```
(C3) eq2:ev(eq1,x=X*cos(phi)-Y*sin(phi),y=X*sin(phi)+Y*cos(phi),expand)$
```

Nous sortons maintenant le coefficient de XY :

```
(C4) cXY:coeff(coeff(eq2,X),Y);
```

(D4) $$24\sin^2\varphi - 14\cos\varphi\sin\varphi - 24\cos^2\varphi$$

On simplifie un peu.

```
(C5) cXY:trigrat(cXY);
```

(D5) $$-7\sin(2\varphi)-24\cos(2\varphi)$$

Pour trouver φ de sorte que ce coefficient soit nul, nous allons introduire la tangente.

```
(C6) cXY:ev(cXY/cos(2*phi),sin(2*phi)=tan(2*phi)*cos(2*phi),factor);
```

(D6) $$-(7\tan(2\varphi)+24)$$

Il ne reste qu'à résoudre...

```
(C7) solphi:solve(cXY=0,phi);
```

(D7) $$\left[\varphi=-\frac{\operatorname{Arctan}\left(\frac{24}{7}\right)}{2}\right]$$

```
(C8) phi:part(solphi,1,2)$
```

On calcule le cosinus de φ, pour cela il faut utiliser une petite astuce, on passe par le carré du cosinus que l'instruction `trigrat` gère bien parce qu'il s'agit d'un angle moitié...

```
(C9) cphi:sqrt(trigrat(cos(phi)^2));
```

(D9) $$\frac{4}{5}$$

Pour le sinus on procède de la même façon mais en prenant en charge le signe, ce qui n'était pas nécessaire pour le cosinus puisque l'angle φ est compris entre $-\frac{\pi}{4}$ et $\frac{\pi}{4}$.

```
(C10) sphi:signum(phi)*sqrt(trigrat(sin(phi)^2));
```

(D10) $$-\frac{3}{5}$$

```
(C11) eq2:ev(eq1,x=X*cphi-Y*sphi,y=X*sphi+Y*cphi,expand)$
```

Nous pouvons donc obtenir maintenant l'équation de $\mathscr{C}$ dans le nouveau repère.

```
(C12) eq2=0;
```

(D12) $$5\,Y+25\,X^2+40\,X=0$$

```
/* Réduction des courbes du second degré (phase 1 : élimination du terme */
/* rectangle)                                                               */
PARTSWITCH:true;
reduction_1(expr) := block(
    /* les variables locales                                                */
    [e,d,p,c,s],
    e:ev(expr,x=X*cos(phi)-Y*sin(phi),y=X*sin(phi)+Y*cos(phi),expand),
    d:coeff(coeff(e,X),Y),
    d:trigrat(d),
    d:ev(d/cos(2*phi),sin(2*phi)=tan(2*phi)*cos(2*phi),factor),
    p:part(solve(d=0,phi),1,2),
    /* Prise en compte du cas particulier ou sin(2phi) n'apparaît pas    */
    /* dans le coefficient de XY - il n'y a pas de solution à l'équation */
    /* précédente. La valeur qui convient pour phi est alors pi/4.       */
    p:if p=end then %pi/4 else p,
    c:sqrt(trigrat(cos(p)^2)),
    s:signum(p)*sqrt(trigrat(sin(p)^2)),
    ev(expr,x=X*c-Y*s,y=X*s+Y*c,expand)
    );
```

maxima >>

```
(C2) load("reduction_2nddegre_1.mc")$
```

```
(C3) e:16*x^2-24*x*y+9*y^2+35*x-20*y$
```

```
(C4) e=0;
```

(D4) $$9\,y^2-24\,x\,y-20\,y+16\,x^2+35\,x=0$$

```
(C5) reduction_1(e)=0;
```

(D5) $$5\,Y+25\,X^2+40\,X=0$$

```
(C6) e:x^2+x*y+y^2-1$
```

```
(C7) e=0;
```

(D7) $$y^2+x\,y+x^2-1=0$$

```
(C8) reduction_1(e)=0;
```

(D8) $$\frac{Y^2}{2}+\frac{3\,X^2}{2}-1=0$$

```
(C9) e:x^2+2*y^2+4*sqrt(3)*x*y +x+sqrt(3)*y+1$

(C10) e=0;
```

(D10) $$2\,y^2+4\,\sqrt{3}\,x\,y+\sqrt{3}\,y+x^2+x+1=0$$

```
(C11) reduction_1(e)=0;
```

(D11) $$5\,Y^2+\frac{3\,\sqrt{3}\,Y}{\sqrt{7}}-2\,X^2-\frac{X}{\sqrt{7}}+1=0$$

```
(C12) e:5*x^2+7*y^2+2*sqrt(3)*x*y-(10+2*sqrt(3))*x-(14+2*sqrt(3))*y-4+2*sqrt(3)$

(C13) e=0;
```

(D13) $$7\,y^2+2\,\sqrt{3}\,x\,y+\left(-2\,\sqrt{3}-14\right)\,y+5\,x^2+\left(-2\,\sqrt{3}-10\right)\,x+2\,\sqrt{3}-4=0$$

```
(C14) reduction_1(e)=0;
```

(D14) $$8\,Y^2-8\,\sqrt{3}\,Y-8\,Y+4\,X^2-4\,\sqrt{3}\,X+4\,X+2\,\sqrt{3}-4=0$$

```
(C15) e:x^2+y^2+2*x-3*y+1$

(C16) e=0;
```

(D16) $$y^2-3\,y+x^2+2\,x+1=0$$

```
(C17) reduction_1(e)=0;
```

(D17) $$Y^2-\frac{5\,Y}{\sqrt{2}}+X^2-\frac{X}{\sqrt{2}}+1=0$$

2.2 Translation du repère

Cette opération est destinée à éliminer l'éventuel terme de degré 1 en présence d'un terme de degré 2 pour une variable donnée. *Développement à suivre ....*